



```


.control
run
set pensize = 2
plot vin bp hp lp title FIN_equals_1
alter VFIN dc = 1
run
plot vin bp hp lp title FIN_equals_.7
alter VFIN dc = 1.5
run
plot vin bp hp lp title FIN_equals_1.5
.endc

```


Initially, **VFIN** set the input to be 1Hz.



At 1Hz, the Bandpass voltage should equal the input voltage exactly. And both the Lowpass and Highpass will be equal in magnitude, but not in phase.




At 0.7Hz, all the outputs are unequal in size. The Q has been set to one. So the lowpass is slightly higher than VIN. And the Highpass has the lowest magnitude.



At 1.5Hz, the Highpass has the highest magnitude and the low pass the lowest.

=====**Full Netlist For Copy Paste**=====

DC_Controlled_StateVariable_Transient



=====**Create Signal**=====


```

VT      VT      0      DC      0      PWL( 0 0 100 100)
BVIN    VIN     0      V       =     1*sin(6.283*V(FIN)*V(VT))

VFIN    FIN     0      DC      .7
VFC     FC      0      DC      1
VK      K       0      DC      1
VA1     A1      0      DC      1
VA2     A2      0      DC      1
VB0     B0      0      DC      1
VB1     B1      0      DC      -1u
VB2     B2      0      DC      1u
XStates VIN    FC    K      A1  A2  B0  B1  B2  VOUT HP  BP  LP  StateVS
*TRAN   TSTEP  TSTOP  TSTART TMAX ?UIC?
.tran   30u    10     8      30u   UIC
.control
run
set     pensize = 2
plot   vin  bp  hp  lp  title FIN_equals_.7
alter  VFIN      dc = 1
run
plot   vin  bp  hp  lp  title FIN_equals_1
alter  VFIN      dc = 1.5
run
plot   vin  bp  hp  lp  title FIN_equals_1.5
.endc


```

-----StateVariable_Cell_S-----



```
.SUBCKT StateVS VIN FC K A1 A2 B0 B1 B2 VOUT HP BP LP
R1 HP 0 1
R2 OUT 0 1
Bgmin HP 0 I = -V(VIN)*V(K)*1
Bgma1 HP 0 I = V(LP)*V(A1)
Bgma2 HP 0 I = (V(BP))*V(A2)
Bgmb0 OUT 0 I = -V(LP)*V(B0)
Bgmb1 OUT 0 I = -V(BP)*V(B1)
Bgmb2 OUT 0 I = -V(HP)*V(B2)
XS1block HP BP FC Sblock
XS2block BP LP FC Sblock
BOUT VOUT 0 V = V(OUT)
.ENDS StateVS
```

-----S_BLOCK-----



```
For FC = 1Hz
RS = 1 Ohm CS=1uF/2Pi
Xc = 1 Ohm
.SUBCKT Sblock VIN OUT FC
Bbuf IN 0 V = -V(VIN)
BRS IN VNEG I = (V(IN)-V(VNEG))/V(FC)
Cs VNEG OUT .159
BSOUT OUT 0 V = -V(VNEG)*300
.ENDS Sblock
.end
```

6.7.11_10.22AM
dsauersanjose@aol.com
Don Sauer